# Copyright (C) 2018 Google Inc.
#
# Licensed under the Apache License, Version 2.0 (the "License");
# you may not use this file except in compliance with the License.
# You may obtain a copy of the License at
#
#      http://www.apache.org/licenses/LICENSE-2.0
#
# Unless required by applicable law or agreed to in writing, software
# distributed under the License is distributed on an "AS IS" BASIS,
# WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
# See the License for the specific language governing permissions and
# limitations under the License.

java_library(
    name = "platform",
    srcs = select({
        "//tools/build:linux": glob(["linux/**/*.java"]),
        "//tools/build:darwin": glob(["osx/**/*.java"]),
        "//tools/build:windows": glob(["windows/**/*.java"]),
    }),
    resource_strip_prefix = "gapic/src/platform/",
    resources = select({
        "//tools/build:linux": [":liblinux_glcanvas.so"],
        "//conditions:default": [],
    }),
    visibility = ["//visibility:public"],
    deps = [
        "@gapic_third_party//:lwjgl",
        "@gapic_third_party//:swt",
    ],
)

cc_binary(
    name = "liblinux_glcanvas.so",
    srcs = [
        "jni.h",
        "jni_md.h",
        "linux/glcanvas.cc",
    ],
    includes = ["."],
    linkopts = [
        "-lGL",
        "-lX11",
    ],
    linkshared = 1,
)

genrule(
    name = "copy_jni.h",
    srcs = ["@bazel_tools//tools/jdk:jni_header"],
    outs = ["jni.h"],
    cmd = "cp -f $< $@",
)

genrule(
    name = "copy_jni_md.h",
    srcs = ["@bazel_tools//tools/jdk:jni_md_header-linux"],
    outs = ["jni_md.h"],
    cmd = "cp -f $< $@",
)
